
Introduction

This document describes use of the SPSS data file which can be used for attaching CAMSIS scale scores to the occupations of individuals in the CHER datasets. Score matching can be achieved by linking the CHER datafiles with a CAMSIS ‘index’ file which includes the appropriate CAMSIS scores alongside the occupational unit. The relevant files are currently located in : 

T:\IRISS\Visitors docs\camsis\cher_camsis\*.* 

This report accompanies a longer report in which the derivation of CAMSIS scale scores relevant to the CHER data, and the properties of those scores, are discussed. 

The CAMSIS approach

The CAMSIS (‘Cambridge Social Interaction and Stratification Scales’) approach involves assigning scores to occupational units which are said to reflect the locations of occupations within the social structure of a stratification order. The scores are assigned to occupations on the basis of information on the social interaction patterns typically exhibited by occupational unit holders, under the theoretical assumption that social interaction patterns inherently reflect social stratification structures. In practice CAMSIS scores are scaled to have a mean of 50 and fit within the range (1,99), where higher scores are assigned to occupations whose social interaction patterns suggest that they are associated with positions of greater social advantage. Importantly, CAMSIS scores are derived as relative to the national context and gender group – separate CAMSIS scores apply to occupations in each different country, and to the distinguishable occupational structure of men and women. This property distinguishes the CAMSIS scores from many other representations of occupational stratification, which are typically imposed as equivalent between genders and nations; the equivalent relative nature, but national and gender context, of the CAMSIS scores can be argued to be more appropriate tool for conducting cross-nationally comparative research. 

A full introduction and description of the CAMSIS approach, the derivation of CAMSIS scores, and their properties, can be accessed through the CAMSIS project wepbages, 

http://www.cf.ac.uk/socsi/CAMSIS/

The CAMSIS datasets associated with the CHER studies, then, enable access to CAMSIS scores assigned to the occupations of CHER study members. The scores can be matched with CHER occupations conditional on information on the gender and nationality of their incumbents. This is achieved through matching with a data file which links CAMSIS scores with occupational units in each country and gender group. Most of the components of the dataset were created during an IRISS visit to CEPS in September 02, though later updates may be adopted. For information, the CAMSIS dataset also contains variables which indicate the scores assigned to occupations under two alternative stratification schemes, the ISEI (International Socio-Economic Status Index) and SIOPS (Standard International Occupational Prestige Scores), as obtained by translating ISCO codes under macros provided by Ganzeboom and Treiman (see http://www.fss.uu.nl/soc/hg/isko88/).

Linking CHER datasets with available CAMSIS measures

It is possible to associate CAMSIS scores with the occupations of individuals in those CHER studies which include occupational unit details. For almost all countries the harmonised individual occupational unit variable in the CHER datasets is ISCO-88(com). This is recorded in the CHER person file variable “p{xx}l04”, aggregated to the 2-digit level in all but one case. This level of aggregation is less desirable from the perspective of the CAMSIS approach, where maximal occupational detail is preferred, but the multiple categories should nevertheless provide an adequate basis for representing occupational differentiation through social association. However not all of the component countries from the CHER datasets currently record individual occupational unit information in this variable. 

In the CAMSIS project as coordinated at Cardiff University, detailed CAMSIS scores are derived for occupations in a number of countries, based upon extensive large scale datasources such as national census’s and aggregations of survey datasets. For some countries from CHER, such completed CAMSIS scores are available to be linked with the CHER data, but for other countries this is not at present the case. Table 1 illustrates the threefold division in the coverage of  CHER datasets by CAMSIS scores at time of writing (1.10.02). For four countries, we are able to associate scores from completed CAMSIS versions (derived themselves from census or large scale survey datasets). For a further seven countries we have derived ‘working’ CAMSIS versions at the ISCO-88 2-digit level using the CHER databases themselves (see further notes in the associated report). Lastly for eight countries there is no adequate occupational detail recorded in the CHER datasets and no CAMSIS measures are applied
. 

Table 1: CHER datasets by CAMSIS coverage, current CHER files of 2.10.02







Country (numb*) 
Occupational data present
Associated CAMSIS scores









Germany (3)
90,91,93,94,95,96,97,98,99,00
Final version ISCO-68 3-digit

Switzerland (9)
99,00
Final version ISCO-88 2-digit

United Kingdom (10)
91,93,94,95,96,97,98,99,00
``

Ireland (17)
94,95,96,97,98
``









Belgium (1)
94,95,96,97,98
Derived version ISCO-88 2-digit

Hungary (4)
93,94,95,96,97,98
``

Luxembourg (6)
95,96,97,98,99,00
``

Poland (8)
97,98,99,00
``

Denmark (13)
94,95,96,97,98
``

France (15)
94,95,96,97,98
``

Portugal (18)
94,95,96,97,98
``









Canada (2)

(national CAMSIS expected)

Italy (5)



The Netherlands (7)

(national CAMSIS expected)

United States (11)

(national CAMSIS completed)

Austria (12)

(national CAMSIS expected)

Finland (14)

(national CAMSIS expected)

Greece (16)



Spain (19)

(national CAMSIS expected)





* Indicates country number as used in the harmonised CHER dataset

Using the CAMSIS files

To reiterative, the relevant CAMSIS scores follow a different distribution for each country and, within countries, by gender groups. The supplied SPSS ‘index’ file therefore connects CAMSIS score values to occupations conditional upon country and gender group. The basic set of SPSS commands to utilise this file are relatively simple and given below (primarily in section (iii). However the different occupational base information for, specifically. the German CHER files adds a further complication. In this case it is necessary to translate the German CHER ISCO-68 information into an appropriate ISCO-88 format before proceeding. This may be achieved by following macros provided as ‘include’ files as downloaded from a webpage mainted by Harry Ganzeboom as part of the ISMF project
. Such a file has been placed in a subfolder accessed from the folder locating the primary CAMSIS CHER datasets.

Once this translation for German cases is done, the suggested method of linking the CAMSIS scores with a CHER database involves utilising SPSS’s ‘match files’ commands. The text below shows SPSS syntax which can be used for this purpose (it can also be found in an example syntax file located in the same CAMSIS CHER folder. The variable names used on the CAMSIS index file are ‘isco’, ‘gender’,  ‘country’, ‘cs’, ‘isei’ and ‘siops’. These reflect the ISCO (two or three digit) unit, the gender and CHER country number relevant, and the CAMSIS, then ISEI and SIOPS, scores respectively. Typically users may prefer to use alternative naming conventions, which can be achieved by renaming variables as in the example below. 



** Example SPSS syntax, methods for matching CHER datasets to CAMSIS scores via file  'camsis_all.sav' .

** Example: access a CHER file from 1997 at the person file level (plus a few other vars)  . 

**i)  Matching requires gender and individual level occupational info so at a 

*      minimum we need to match person file with meta file. 

get file="chermeta.sav"      /keep=pid country pd02 /rename(pd02=gender) .

sort cases by country pid. 

fre var=country.

* 370732 cases anywhere here.

sav out="mtch1.sav" . 

get file="cher97p.sav"  /keep=pid country year p97l04  p97e01 p97e02 p97h03 p97i02 .

sort cases by country pid. 

match files file=* /in=persf /file="mtch1.sav" /by=country pid. 

fre var=year persf . 

select if (persf=1).

descriptives var=all.

* (we've retained 172099 cases from 1997 person files) . 

** ii) Check and prepare the occupational info; 

*         for most countries we have the appropriate 2 digit ISCO88,

*         but for Germany the current format is 3 digit ISCO68, which we need to adjust.    

recode p97l04 (missing,sysmis, lo thru -1=-999).

missing values p97l04 (-999).

means tables=p97l04 by country /cells=count mean stddev min max .

* First compute an index variable based on the appropriate CHER occupational variable, 

* but adjusting specifically the German ISCO68 variable into an appropriate ISCO88 translation. 

compute occ=-999.

if (p97l04 ge 0) occ=p97l04.

descriptives var=occ. 

* (below file allows German 68 to 88 translations).

include file="\cher_camsis\resources\hg68to883.sps".

fre var=country.

temp.

select if (country=3 | country=51 | country=61).

cro occ by country . 

do if ( (country=3 | country=51 | country=61) & occ gt 0).

* most are 2 or 3 digit 68 but macro translates 4 digit, so need to multiply by 10.

compute occ=-(p97l04*10).

* below to cater for a few possible one digit units (for 68 major groups).

if (p97l04 ge 1 & p97l04 le 9) occ=-(p97l04* 1000).

is68t88 occ88={occ}.

* We've now converted to ISCO88, but in terms of 4 digit, and better here to truncate to 3 .

* this requires 3 sequential steps to also deal correctly with subgroup info .

if ( ((occ/100) - trunc(occ/100) = 0 )) occ=trunc(occ/100).

if ( ((occ/1000) - trunc(occ/1000) = 0 ) & occ ne 1000) occ=trunc(occ/1000).

if (occ ge 1000 & occ le 9999) occ=trunc(occ/10).

end if.

execute.

* End of German special treatment (note that there is no provision for armed forces in German

* translation, any isco68's in armed forces will just be ignored). 

missing values occ (-999).

* In some countries minor groups 11 and 12 are merged.

recode occ (11.12=11).

* add value labels for occ :.

include file="\cher_camsis\resources\isco88comlabelsv1.sps".

istilab occ={occ}.

temp.

select if (country ne  3 & country ne 51 & country ne 61).

fre var=occ. 

temp.

select if (country=3 | country=51 | country=61).

fre var=occ. 

* (the two freqs show the two different levels of detail).

** iii) Now match the (harmonised) occupation index varaible with CAMSIS file. 

* is is also desirable to have any relevant echp datasets coded to their main country number, so use ‘country2’.

compute country2=country. 

recode country2 (51,61=3) (57,67=10) (71=1) (72=6) .

sort cases by country2 occ gender. 

sav out="mtch1.sav".

get file="\camsis_all.sav".

rename var (country=country2) (isco=occ) (cs=cs97) (isei=isei97) (siops=siops97) . 

sort cases by country2 occ gender.

match files file="mtch1.sav" /table=* /by=country2 occ gender.

recode cs97 isei97 siops97 (missing,sysmis, lo thru 0=-999).

missing values cs97 isei97 siops97 (-999).

descriptives var=p97l04 occ cs97 isei97 siops97. 

** we have 49724 people in valid occs on the 1997 file; 

*      49369 of them we were able to give cs scores, and 49326 get isei's and siops. 

*      (the difference is mainly due to absence of armed forces info for isei and siops, 

*      and the non-coverage of some isco occs by any scheme). 

** iv) Lastly some checks on the distributions : .

graph /scatterplot=cs97 with isei97. 

graph /scatterplot=cs97 with siops97. 

* These just suggest that all counties have camsis score of same nature as isei or siops,

*  which is fairly similar to ISEI or SIOPS .

* across countries.

temp.

select if (gender=1).

graph /histogram=cs97.

temp.

select if (gender=1).

descriptives var=cs97 isei97 siops97 /statistics=mean min max stddev skewness kurtosis . 

temp. 

select if (gender=2).

graph /histogram=cs97.

temp.

select if (gender=2).

descriptives var=cs97 isei97 siops97 /statistics=mean min max stddev skewness kurtosis . 

* Expect within country differences to be greater by isei and siops than cs98.

temp. 

select if (gender=1).

means tables=cs97 isei97 siops97 by country 

            /cells=mean stddev count min max /statistics=anova.

temp. 

select if (gender=2).

means tables=cs97 isei97 siops97 by country 

            /cells=mean stddev count min max /statistics=anova.

* (here, least between country variation is in CAMSIS, most is in ISEI).

********************************************************.





Additional notes on the usage of the file ‘camsis_all.sav’ 

i) For the bulk of the CHER countries covered, we have assigned CAMSIS scores at the two digit ISCO sub-major group level, and also at the single digit ISCO minor-group level. We have used the convention however that the major group unit values equal 10 times the major group unit, so for instance the scores for major group 1 and 2 are associated with units 10 and 20. The reason for this practice is the confusion inherent to the ISCO-88 schema between occupation units located in major groups 0 and 1. For instance, with no prior specification it is impossible to distinguish between an indication of major group 1 (Managers) or sub-major group 1 (armed forces). By making this specification, we adopt specifically the following relevant notation : 0 = “Major group 0, armed forces” 1 = Sub-major group 1 (armed forces); 10 = “Major group 1, managers”. Users may also note that several CAMSIS scores are marked as missing for the armed forces groups, this occurring when no cases were recorded as such in the CHER datasets.  

ii) It may be expected that those CAMSIS versions derived as ‘working versions’ using the sparser CHER datasets themselves, will be less stable representations of the occupational structure for the given country, in comparison to those CAMSIS scores derived from larger scale data resources. 

iii) Due to the relatively limited number of cases available for the CHER based derived CAMSIS scores, none of the CAMSIS versions linked through the CAMSIS_CHER dataset incorporate differences in the employment status positions of occupational holders, although such information is regularly used as part of the CAMSIS scale derivation approach. 

EOF. 

� This table does not include details on the additional datasets included from  four CHER countries which are based upon ECHP samples, though in each case the same opportunities for matching CAMSIS scores apply.


� http://www.fss.uu.nl/soc/hg/isko88/

